```
> restart:
```

MAXWELL's ELECTRODYNAMICS

cyclide1.mws

Updated 02/09/98 -- R. M. Kiehn
Computes:
Maxell-Faraday formulas from a 1-form of Action on 4D space time.
Maxwell Ampere equations assuming Lorentz vacuum constitutive equations
D = epsilon E   B = mu H
Computes Torsion vector = ExA+Bphi, AdotB
computes Spin Vector = A x H + Dphi, AdotD
First Poincare invariant is F^G -A^J = (BdotH-DdotE) -(AdotJ-rhi.phi)
Second Poincare Invariant = F^F = -2EdotB
THE GENERAL PROCEDURE IS:
Given vector potentials A,phi. compute E,B.   compute D,H.
compute J,rho, compute poincare invariants,
compute torsion current and spin current.

(Fundamental Lagrangian is a 3-form. L = A^G +iA^F/Z(hall)??)

Two DIFFERENT topological invariants are possible: torsion and spin (helicity and spin)
Invariance requires that divergence of Spin = 0 and divergence of Torsion = 0.

d(A^F) = F^F   ---- (torsion)

d(A^G)=F^G-A^J   ----(spin)

The Invariance conditions are distinct.

Spin can be conserved, Torsion can be conserved. But these are independent ideas.

```
>  with(liesymm):with(linalg):with(plots):
Warning, new definition for close
Warning, new definition for norm
Warning, new definition for trace
```

```
> setup(x,y,z,t);
```

$$[x, y, z, t]$$

```
> defform(x=0,y=0,z=0,t=0,a=const,b=const,c=const,k=const,mu=const,m=const);
```

$$\mathrm{defform}(x = 0, y = 0, z = 0, t = 0, a = const, b = const, c = const, k = const, \mu = const, m = const)$$

```
> dR:=[d(x),d(y),d(z),d(t)];
```

$$dR := [\mathrm{d}(x), \mathrm{d}(y), \mathrm{d}(z), \mathrm{d}(t)]$$

Specify the four functions that are the covariant components of the Action 1-form.

```
> A1:=Ax(x,y,z,t);
```

```
> A2:=Ay(x,y,z,t);
```

$$A1 := \mathrm{Ax}(x, y, z, t)$$

$$A2 := \mathrm{Ay}(x, y, z, t)$$

```
> A3:=Az(x,y,z,t);
```

$$A3 := \mathrm{Az}(x, y, z, t)$$

```
> A4:=phi(x,y,z,t);
```

$$A4 := \phi(x, y, z, t)$$

Skip the next 3 code lines to use abstract formulas, OTHERWISE enter your own formulas for the vector and scalar potentials. DO NOT CHANGE ANY OF THE OTHER EQUATIONS

****

```
> rr:=x^2+y^2+z^2-(c*t)^2;
```

$$rr := x^2 + y^2 + z^2 - c^2 t^2$$

The Phi function must satisfy the scalar Helmholtz equation

Skip th next line for abstract formulas

```
> AA:=evalm([y,-x,c*t]/rr^2);
```

$$AA := \left[\frac{y}{(x^2 + y^2 + z^2 - c^2 t^2)^2}, -\frac{x}{(x^2 + y^2 + z^2 - c^2 t^2)^2}, \frac{c\, t}{(x^2 + y^2 + z^2 - c^2 t^2)^2} \right]$$

```
> A4:=z*c/(rr^2);
```

$$A4 := \frac{z\, c}{(x^2 + y^2 + z^2 - c^2 t^2)^2}$$

****

```
> A1:=factor(simplify(AA[1]));A2:=factor(simplify(AA[2]));A3:=factor(simplify(AA[3
  ]));div4A:=factor(simplify(diverge(AA,[x,y,z])+diff(A4,t)/c^2));
```

$$A1 := \frac{y}{(x^2 + y^2 + z^2 - c^2 t^2)^2}$$

$$A2 := -\frac{x}{(x^2 + y^2 + z^2 - c^2 t^2)^2}$$

$$A3 := \frac{c\, t}{(x^2 + y^2 + z^2 - c^2 t^2)^2}$$

$$div4A := 0$$

****

```
> Action:=wcollect(A1*d(x)+A2*d(y)+A3*d(z)-A4*d(t));
```

$$Action := \frac{y\, \mathrm{d}(x)}{\%1^2} - \frac{x\, \mathrm{d}(y)}{\%1^2} + \frac{c\, t\, \mathrm{d}(z)}{\%1^2} - \frac{z\, c\, \mathrm{d}(t)}{\%1^2}$$

$$\%1 := x^2 + y^2 + z^2 - c^2 t^2$$

The 2-form of Field Intensities obtained from the 1-form of Action.

```
> F:=wcollect(d(Action));
```

$$F := \left(-4 \frac{c^2 t\, x}{\%1^3} - 4 \frac{z\, c\, y}{\%1^3} \right) ((\mathrm{d}(t))\ \&\hat{}\ (\mathrm{d}(y))) + \left(4 \frac{x\, c\, t}{\%1^3} - 4 \frac{z\, y}{\%1^3} \right) ((\mathrm{d}(z))\ \&\hat{}\ (\mathrm{d}(x)))$$
$$+ \left(4 \frac{c^2 t\, y}{\%1^3} - 4 \frac{z\, c\, x}{\%1^3} \right) ((\mathrm{d}(t))\ \&\hat{}\ (\mathrm{d}(x))) + \left(-\frac{-3 y^2 + x^2 + z^2 - c^2 t^2}{\%1^3} + \frac{3 x^2 - y^2 - z^2 + c^2 t^2}{\%1^3} \right) ((\mathrm{d}(x))\ \&\hat{}\ (\mathrm{d}(y)))$$

$$+\left(4\frac{z\,x}{\%1^3}+4\frac{y\,c\,t}{\%1^3}\right)((\mathrm{d}(z))\ \&\wedge\ (\mathrm{d}(y)))+\left(\frac{c\,(x^2+y^2-3\,z^2-c^2\,t^2)}{\%1^3}+\frac{c\,(3\,c^2\,t^2+x^2+y^2+z^2)}{\%1^3}\right)((\mathrm{d}(t))\ \&\wedge\ (\mathrm{d}(z)))$$

$$\%1 := x^2+y^2+z^2-c^2\,t^2$$

The components of the Vector potential in engineering format.

```
> A:=[A1,A2,A3]:
>
```

Magnetic field intensity

```
> B:=curl(A,[x,y,z]):B1:=factor(simplify(B[1]));B2:=factor(simplify(B[2]));B3:=factor(simplify(B[3]));BdotB:=factor(simplify(innerprod(B,B)));
```

$$B1 := -4\,\frac{y\,c\,t+z\,x}{(x^2+y^2+z^2-c^2\,t^2)^3}$$

$$B2 := 4\,\frac{x\,c\,t-z\,y}{(x^2+y^2+z^2-c^2\,t^2)^3}$$

$$B3 := 2\,\frac{x^2+y^2-z^2+c^2\,t^2}{(x^2+y^2+z^2-c^2\,t^2)^3}$$

$$BdotB := 4\,\frac{6\,y^2\,c^2\,t^2+2\,z^2\,x^2+6\,x^2\,c^2\,t^2+2\,z^2\,y^2+x^4+2\,x^2\,y^2+y^4+z^4-2\,z^2\,c^2\,t^2+c^4\,t^4}{(x^2+y^2+z^2-c^2\,t^2)^6}$$

The Electric Field Intensity

```
> E:=[-diff(A4,x)-diff(A[1],t),-diff(A4,y)-diff(A[2],t),-diff(A4,z)-diff(A[3],t)]:
E1:=factor(simplify(E[1]));E2:=factor(simplify(E[2]));E3:=factor(simplify(E[3]))
;EdotE:=factor(simplify(innerprod(E,E)));
```

$$E1 := 4\,\frac{c\,(z\,x-y\,c\,t)}{(x^2+y^2+z^2-c^2\,t^2)^3}$$

$$E2 := 4\,\frac{c\,(z\,y+x\,c\,t)}{(x^2+y^2+z^2-c^2\,t^2)^3}$$

$$E3 := -2\,\frac{c\,(x^2+y^2-z^2+c^2\,t^2)}{(x^2+y^2+z^2-c^2\,t^2)^3}$$

$$EdotE := 4\,\frac{c^2\,(6\,y^2\,c^2\,t^2+2\,z^2\,x^2+6\,x^2\,c^2\,t^2+2\,z^2\,y^2+x^4+2\,x^2\,y^2+y^4+z^4-2\,z^2\,c^2\,t^2+c^4\,t^4)}{(x^2+y^2+z^2-c^2\,t^2)^6}$$

Topological Parity (Second Poncare invariant) depends on topology of field intensities.

```
> EdotB:=factor(innerprod(E,B));
```

$$EdotB := -4\,\frac{c}{(x^2+y^2+z^2-c^2\,t^2)^4}$$

The 4 components of the topological Torsion current.

```
> ExA:=crossprod(E,A):Bphi:=factor(simplify([B[1]*A4,B[2]*A4,B[3]*A4])):
> TORS:=evalm(ExA+A4*B):
> AdotB:=factor(inner(A,B));AdotE:=factor(inner(A,E));
```

$$AdotB := -2\,\frac{c\,t}{(x^2+y^2+z^2-c^2\,t^2)^4}$$

$$AdotE := -2\,\frac{c^2\,t\,(3\,y^2+3\,x^2-z^2+c^2\,t^2)}{(x^2+y^2+z^2-c^2\,t^2)^5}$$

AdotB is the classic helicity -- This is not a scalar or a pseudo scalar, but it is the fourth component of a third rank covariant tensor.

```
> TORSION:=[factor(simplify(TORS[1])),factor(simplify(TORS[2])),factor(simplify(TORS[3])),AdotB];
```

$$TORSION := \left[-2\,\frac{c\,x}{\%1^4},\,-2\,\frac{c\,y}{\%1^4},\,-2\,\frac{z\,c}{\%1^4},\,-2\,\frac{c\,t}{\%1^4}\right]$$

$$\%1 := x^2+y^2+z^2-c^2\,t^2$$

Divergence of the Torsion current.

```
> DIVT:=factor(diverge(TORSION,[x,y,z,t]));
```

$$DIVT := 8\,\frac{c}{(x^2+y^2+z^2-c^2\,t^2)^4}$$

The Next step is to compute the Field Excitations assuming a Lorentz vacuum constitutive relation, B = mu H and D = e E. Once having D and H , then compute charge current density from the Maxwell Ampere equations.

The second Poincare Invariant is the related to the difference between the magnetic and the electric energy densities (the lagrangian) minus the interaction between the 4 current and the 4 potential.

First compute the Lagrangian, and along the way the components of the Poynting vector. The results are to within a scalar factor:

```
> EdotE:=innerprod(E,E):BdotB:=innerprod(B,B):ExB:=crossprod(E,B/mu):Lag:=factor(simplify((-epsilon*EdotE+BdotB/mu)));POYX:=factor(simplify(ExB[1]));POYY:=factor(simplify(ExB[2]));POYZ:=factor(simplify(ExB[3]));
```

$$Lag := -4\,\frac{(6\,y^2\,c^2\,t^2+2\,z^2\,x^2+6\,x^2\,c^2\,t^2+2\,z^2\,y^2+x^4+2\,x^2\,y^2+y^4+z^4-2\,z^2\,c^2\,t^2+c^4\,t^4)\,(\varepsilon\,c^2\,\mu-1)}{(x^2+y^2+z^2-c^2\,t^2)^6\,\mu}$$

$$POYX := 16\,\frac{c^2\,(x^2+y^2-z^2+c^2\,t^2)\,x\,t}{(x^2+y^2+z^2-c^2\,t^2)^6\,\mu}$$

$$POYY := 16\,\frac{c^2\,(x^2+y^2-z^2+c^2\,t^2)\,y\,t}{(x^2+y^2+z^2-c^2\,t^2)^6\,\mu}$$

$$POYZ := 32\,\frac{c^2\,z\,t\,(x^2+y^2)}{(x^2+y^2+z^2-c^2\,t^2)^6\,\mu}$$

Now the components of the 4 current:

```
> J:=
  evalm((curl(B,[x,y,z])/mu-epsilon*[diff(E[1],t),diff(E[2],t),diff(E[3],t)])):
> J1:=factor(simplify(J[1],trig));J2:=factor(simplify(J[2]));J3:=factor(simplify(J[3]));rho:=factor(simplify(diverge(E,[x,y,z])));EX:=factor(diff(E[1],x));EY:=factor(diff(E[2],y));EZ:=factor(diff(E[3],z));DDE:=factor(simplify(EX+EY+EZ));
```

$$J1 := 4\,\frac{(x^2\,y+y^3+z^2\,y+5\,y\,c^2\,t^2-6\,x\,c\,t\,z)\,(\varepsilon\,c^2\,\mu-1)}{\mu\,(x^2+y^2+z^2-c^2\,t^2)^4}$$

$$J2 := -4\,\frac{(6\,y\,c\,t\,z+z^2\,x+x^3+y^2\,x+5\,x\,c^2\,t^2)\,(\varepsilon\,c^2\,\mu-1)}{\mu\,(x^2+y^2+z^2-c^2\,t^2)^4}$$

$$J3 := 8\,\frac{c\,t\,(2\,x^2+2\,y^2-z^2+c^2\,t^2)\,(\varepsilon\,c^2\,\mu-1)}{\mu\,(x^2+y^2+z^2-c^2\,t^2)^4}$$

$$\rho := 0$$

$$EX := -4 \frac{c\,(5\,z\,x^2 - z\,y^2 - z^3 + z\,c^2\,t^2 - 6\,c\,t\,y\,x)}{(x^2 + y^2 + z^2 - c^2\,t^2)^4}$$

$$EY := 4 \frac{c\,(-5\,z\,y^2 + z\,x^2 + z^3 - z\,c^2\,t^2 - 6\,c\,t\,y\,x)}{(x^2 + y^2 + z^2 - c^2\,t^2)^4}$$

$$EZ := 8 \frac{c\,z\,(2\,x^2 + 2\,y^2 - z^2 + c^2\,t^2)}{(x^2 + y^2 + z^2 - c^2\,t^2)^4}$$

$$DDE := 0$$

\>

Next compute the Interaction between the 4-current and the 4 potential

```
> INTERACTION:=factor(simplify(innerprod(A,J)-A4*rho));
```

$$INTERACTION := 4 \frac{(9\,y^2\,c^2\,t^2 + 9\,x^2\,c^2\,t^2 - 2\,z^2\,c^2\,t^2 + x^4 + z^2\,x^2 + z^2\,y^2 + 2\,x^2\,y^2 + 2\,c^4\,t^4 + y^4)\,(\varepsilon\,c^2\,\mu - 1)}{(x^2 + y^2 + z^2 - c^2\,t^2)^6\,\mu}$$

Check to see if the fields satisfy the wave equation (requires that the charge current density be zero)

```
> CCB:=curl(curl(B,[x,y,z]),[x,y,z]):DBT:=[factor(simplify(diff(B[1],t))),factor(simplify(diff(B[2],t))),factor(simplify(diff(B[3],t)))]:DDBT:=factor(simplify(diff(DBT[1],t)));
```

$$DDBT := -24 \frac{c^2\,(3\,y\,c\,t\,x^2 + 3\,y^3\,c\,t + 3\,y\,c\,t\,z^2 + 5\,y\,c^3\,t^3 + z\,x^3 + z\,x\,y^2 + z^3\,x + 7\,z\,x\,c^2\,t^2)}{(x^2 + y^2 + z^2 - c^2\,t^2)^5}$$

```
> factor(simplify(CCB[1]));
```

$$24 \frac{3\,y\,c\,t\,x^2 + 3\,y^3\,c\,t + 3\,y\,c\,t\,z^2 + 5\,y\,c^3\,t^3 + z\,x^3 + z\,x\,y^2 + z^3\,x + 7\,z\,x\,c^2\,t^2}{(x^2 + y^2 + z^2 - c^2\,t^2)^5}$$

One factor of the vector wave equation. It should be zero if the solutions are wave functions.

```
> factor(simplify(-CCB[1]-DDBT/c^2));
```

$$0$$

Now evaluate the spin 3-form A^G

This object is not the same as the torsion 3-form, A^F

Often a duality arguement is made to say that they are the same. But this defeats the Lorentz vacuum constitutive constraint. REsults are to within a factor.

```
> Spin:=evalm((crossprod(A,B)/mu+A4*E*epsilon)):spin4:=factor(simplify(epsilon*innerprod(A,E)));spin1R:=factor(simplify(evalc(Re(Spin[1]))));spin1I:=factor(simplify(evalc(Im(Spin[1]))));spin2R:=factor(evalc(Re(simplify(Spin[2]))));spin2I:=factor(evalc(Im(simplify(Spin[2]))));spin3:=factor(simplify(Spin[3]));
>
```

$$spin4 := -2 \frac{\varepsilon\,c^2\,t\,(3\,y^2 + 3\,x^2 - z^2 + c^2\,t^2)}{(x^2 + y^2 + z^2 - c^2\,t^2)^5}$$

$$spin1R := -2 \frac{x^3 + y^2\,x - z^2\,x + 3\,x\,c^2\,t^2 - 2\,y\,c\,t\,z - 2\,z^2\,c^2\,\varepsilon\,\mu\,x + 2\,z\,c^3\,\varepsilon\,\mu\,y\,t}{\mu\,(x^2 + y^2 + z^2 - c^2\,t^2)^5}$$

$$spin1I := 0$$

$$spin2R := -2 \frac{3\,y\,c^2\,t^2 + 2\,x\,c\,t\,z + x^2\,y + y^3 - z^2\,y - 2\,z^2\,c^2\,\varepsilon\,\mu\,y - 2\,z\,c^3\,\varepsilon\,\mu\,x\,t}{\mu\,(x^2 + y^2 + z^2 - c^2\,t^2)^5}$$

$$spin2I := 0$$

$$spin3 := -2\,\frac{z\,(2\,x^2+2\,y^2+\varepsilon\,c^2\,\mu\,x^2+\varepsilon\,c^2\,\mu\,y^2-z^2\,c^2\,\varepsilon\,\mu+\varepsilon\,c^4\,\mu\,t^2)}{\mu\,(x^2+y^2+z^2-c^2\,t^2)^5}$$

Must fix units later. Note that the spin 3-form has zero divergence in the example while the torsion 3-form does not.!

```
> SPINVECTOR:=[spin1R+I*spin1I,spin2R+I*spin2I,spin3,spin4];TORSIONVECTOR:=TORSION
;DIV4SPIN:=factor(diverge(SPINVECTOR,[x,y,z,t]));DIV4TORS:=DIVT;EDOTB:=EdotB;Pow
erz:=POYZ;Interaction:=INTERACTION;Lagrangian:=Lag;Poincare1:=Lagrangian-Interac
tion;Poincare2=EDOTB;
```

$$SPINVECTOR := \left[-2\,\frac{x^3+y^2\,x-z^2\,x+3\,x\,c^2\,t^2-2\,y\,c\,t\,z-2\,z^2\,c^2\,\varepsilon\,\mu\,x+2\,z\,c^3\,\varepsilon\,\mu\,y\,t}{\mu\,\%1^5},\right.$$
$$-2\,\frac{3\,y\,c^2\,t^2+2\,x\,c\,t\,z+x^2\,y+y^3-z^2\,y-2\,z^2\,c^2\,\varepsilon\,\mu\,y-2\,z\,c^3\,\varepsilon\,\mu\,x\,t}{\mu\,\%1^5},$$
$$\left.-2\,\frac{z\,(2\,x^2+2\,y^2+\varepsilon\,c^2\,\mu\,x^2+\varepsilon\,c^2\,\mu\,y^2-z^2\,c^2\,\varepsilon\,\mu+\varepsilon\,c^4\,\mu\,t^2)}{\mu\,\%1^5},\ -2\,\frac{\varepsilon\,c^2\,t\,(3\,y^2+3\,x^2-z^2+c^2\,t^2)}{\%1^5}\right]$$
$$\%1 := x^2+y^2+z^2-c^2\,t^2$$

$$TORSIONVECTOR := \left[-2\,\frac{c\,x}{\%1^4},\ -2\,\frac{c\,y}{\%1^4},\ -2\,\frac{z\,c}{\%1^4},\ -2\,\frac{c\,t}{\%1^4}\right]$$
$$\%1 := x^2+y^2+z^2-c^2\,t^2$$

$$DIV4SPIN := -4\,\frac{(15\,y^2\,c^2\,t^2+15\,x^2\,c^2\,t^2-4\,z^2\,c^2\,t^2+3\,z^2\,y^2+4\,x^2\,y^2+2\,y^4+z^4+2\,x^4+3\,z^2\,x^2+3\,c^4\,t^4)\,(\varepsilon\,c^2\,\mu-1)}{\mu\,(x^2+y^2+z^2-c^2\,t^2)^6}$$

$$DIV4TORS := 8\,\frac{c}{(x^2+y^2+z^2-c^2\,t^2)^4}$$

$$EDOTB := -4\,\frac{c}{(x^2+y^2+z^2-c^2\,t^2)^4}$$

$$Powerz := 32\,\frac{c^2\,z\,t\,(x^2+y^2)}{(x^2+y^2+z^2-c^2\,t^2)^6\,\mu}$$

$$Interaction := 4\,\frac{(9\,y^2\,c^2\,t^2+9\,x^2\,c^2\,t^2-2\,z^2\,c^2\,t^2+x^4+z^2\,x^2+z^2\,y^2+2\,x^2\,y^2+2\,c^4\,t^4+y^4)\,(\varepsilon\,c^2\,\mu-1)}{(x^2+y^2+z^2-c^2\,t^2)^6\,\mu}$$

$$Lagrangian := -4\,\frac{(6\,y^2\,c^2\,t^2+2\,z^2\,x^2+6\,x^2\,c^2\,t^2+2\,z^2\,y^2+x^4+2\,x^2\,y^2+y^4+z^4-2\,z^2\,c^2\,t^2+c^4\,t^4)\,(\varepsilon\,c^2\,\mu-1)}{(x^2+y^2+z^2-c^2\,t^2)^6\,\mu}$$

$$Poincare1 := -4\,\frac{(6\,y^2\,c^2\,t^2+2\,z^2\,x^2+6\,x^2\,c^2\,t^2+2\,z^2\,y^2+x^4+2\,x^2\,y^2+y^4+z^4-2\,z^2\,c^2\,t^2+c^4\,t^4)\,(\varepsilon\,c^2\,\mu-1)}{(x^2+y^2+z^2-c^2\,t^2)^6\,\mu}$$
$$-\,4\,\frac{(9\,y^2\,c^2\,t^2+9\,x^2\,c^2\,t^2-2\,z^2\,c^2\,t^2+x^4+z^2\,x^2+z^2\,y^2+2\,x^2\,y^2+2\,c^4\,t^4+y^4)\,(\varepsilon\,c^2\,\mu-1)}{(x^2+y^2+z^2-c^2\,t^2)^6\,\mu}$$

$$Poincare2 = -4\,\frac{c}{(x^2+y^2+z^2-c^2\,t^2)^4}$$

Therefore if the Spin 3-form has a zero divergence, it has an evolutionary invariant period integral!!!!.
But if the Spin 3-form is non-zero, then its divergence is equal to the First Poincare Invariant =F^G - A^J.
In the example above, the Spin vector has zero divergence and hence is a conserved quantity if alpha +beta = plus or minus 2.
While the divergence of the torsion vector is not zero unless alpha+beta is zero.

In the exampe, the spin and the torsion vector are conjugate in a sense (is this an artifact of a Clifford algebra), first two components (x and y) are of opposite sign. The example shows the distinct difference between Spin and Torsion.
The Poincare invariants are Divspin and Divtors. They are not both zero in the example. Spin can be conserved , but torsion is not, or visa versa.

```
> CHECKSUM:=factor(DIV4SPIN-Lagrangian+Interaction);
```

$$CHECKSUM := 0$$

The CHECKSUM must vanish if the computations are correct.
****

Next compute the product of the Torsion vector with the 1-form of vector potential. It should vanish if the computations are correct..

```
> IEIRR:=factor(TORSION[1]*A[1]+TORSION[2]*A[2]+TORSION[3]*A[3]-AdotB*A4);
```

$$IEIRR := 0$$

It is remarkable that in the example above if the substitution t => -t is made in the equations for the potentials, then the solutions do not satisfy the vacuum conditions of zero charge current densities.
In other words the example solution is not time reversal invariant.